Paper 169-2012

Using SAS® ODS Report Writing Interface to Create Clinical Study Reports

Rui Duan and Daniel Feaster, University of Miami, Miami, FL, USA

ABSTRACT

Summary tables and listings are the most common components in clinical study reports. Different tables and listings in one study usually share a similar layout. Clinical study reports are often converted to Rich Text Format (RTF), and then distributed and viewed as Microsoft Word files. Different methods have been developed to automate the creation of summary tables and listings. This paper demonstrates a new approach to creating clinical study reports using the SAS ODS Report Writing Interface.

INTRODUCTION

In clinical studies, tables and listings are periodically required for Data Safety Monitoring Board (DSMB) meetings and for final study reports. Different tables and listings in one study usually share a similar layout. Frequently a listing can be viewed as a table with a relatively less dynamic layout. No matter what software is used to generate the tables, after the table creation, the clinical reports are often converted in RTF file and then distributed and viewed as Microsoft Word files. In the final Word files a single table can be in just one page or across multiple pages. Table generation within SAS can be accomplished through multiple procedures such as PROC REPORT, PROC TABULATE and PROC PRINT. With the power of ODS, tables generated from these procedures can be customized to an extent, however frequently the tables in clinical studies need to be so highly customized that the standard SAS procedures cannot meet the requirements. In these cases SAS programmers always count on the DATA _NULL_ programs. Through the power of conditional logic, by processing, array processing and all SAS functions, the DATA _NULL_ programs enable programmers the maximum flexibility and full control of data placement in a table. To have the greatest power of table creation, the DATA _NULL_ programs do require substantially more programming efforts. Sometimes, it is quite difficult to do the debugging, maintenance and modification. So is there a language that is blended with user-friendly coding style and the capability to handle complex table layout within SAS platform? Fortunately, with the updated ODS Report Writing Interface, the answer is yes. The ODS Report Writing Interface is an object-oriented language that provides you with flexibility and control so that even the most rigid reporting requirements can be met with ease. Though it was designed to span the DATA _NULL_ reporting ability to handling non-tabular formats, with its object-oriented programming features, it can be a great choice in clinical table creation.

A clinical report table is usually composed of five sections in one page: page header, table title, table header, table body, table footnote and perhaps a page footer as the sixth section. From the perspective of a SAS ODS Report Writing program, these sections are different objects that can be created using a handful of methods that the interface offers. Between the different objects, conditional logic and by processing in a DATA _null_ program are used to connect the objects and control the data placement.

GENFERATE SUMMARY TABLE

In this example, we generate a table that fills data in the following table shell (Table 1.0). The source dataset is hypothetical and has been shaped and formatted using a macro system to have a layout close to the table shell.

page1 of 2

Study XXXXXX
Table 1.0
Demographics by Treatment, overall[1]

	Treatment A	Treatment B	Treatment AB	Overall	p value[2]
Age					.xxx
N	xxx	xxx	xxx	xxx	
Mean(std)	xx(xx)	xx(xx)	xx(xx)	xx(xx)	
Minimum, Maximum	xx,xx	xx,xx	xx,xx	xx,xx	
Gender					.xxx
Male	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Female	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Ethnicity[3]					.xxx
Hispanic	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Black	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
White	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Other	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Education					.xxx
8th grade or less	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Some high school, no diploma	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
High school diploma or GED	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Some college or technical training	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
College degree (such as BA, BS)	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Any post college education	x/n(%)	x/n(%)	x/n(%)	x/n(%)	

[1]Only randomized participants were included.
[2]Fisher exact test was used for Education.l
[3]If reported Hispanic=Yes then Ethnicity=Hispanic.
Generated from G:\SAS SUGI\demo_treat.sas on November 13, 2011 at 8:13 PM (Cont'd)

page2 of 2

Study XXXXXX

Table 1.0

Demographics by Treatment, overall[1]

	Treatment A	Treatment B	Treatment AB	Overall	p value[2]
Marriage Status					.xxx
Married	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Living with partner/cohabitating	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Widowed	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Separated/divorced	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Single	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
Annual Income					.xxx
Missing (no report)	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
$0 to $5000	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
$5,001 to $10,000	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
$10,001 to $30,000	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
$30,001 to $40,000	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
$40,001 to $50,000	x/n(%)	x/n(%)	x/n(%)	x/n(%)	
More than $50,000	x/n(%)	x/n(%)	x/n(%)	x/n(%)	

[1]Only randomized participants were included.

[2]Fisher exact test was used.

[3]If reported Hispanic=Yes then in ethnicity=Hispanic.

Generated from G:\SAS SUGI\demo_treat.sas on November 13, 2011 at 8:13 PM

Display 1. Table Shell of Summary Table

To duplicate the table shell with data values, we need to create a two page table having six columns. The variable name and p value need to be put in the same line and the categories within the same variable need to be indented. Three superscripts need to be placed, one at the end of the third line of the table title, one behind column header of "p value" and the last behind the variable "Ethnicity". The three super scripts are described by three lines of footnotes which also must be situated at the bottom of the table. Another line of text about source SAS code used, date and time of creation is also created and attached with a "(Con't)" at the end to indicate the continuation of this table except for the last page. Using the code in appendix 1, we can generate Table 1.0 that has exactly the layout as the table shell. For the syntax of ODS writing Interface, please see Daniel O'Connor's papers listed in the reference.

page1 of 2

Study XXXXXX

Table 1.0

Demographics by Treatment, overall[1]

	Treatment A (N=463)	Treatment B (N=280)	Treatment AB (N=257)	Overall (N=1000)	p value[2]
Age					.242
N	463	280	257	1000	
Mean(std)	49.7(6.8)	49.1(6.7)	50.0(6.7)	49.6(6.8)	
Median(p25, p75)	49.7(45.4,53.9)	49.2(44.4,53.7)	49.4(45.1,55.1)	49.6(45.0,54.2)	
Minimum, Maximum	26.0, 70.1	23.1, 66.3	30.1, 65.8	23.1, 70.1	
Gender					.049
Male	406/463(87.7%)	236/280(84.3%)	211/257(82.1%)	853/1000(85.3%)	
Female	57/463(12.3%)	44/280(15.7%)	46/257(17.9%)	147/1000(14.7%)	
Ethnicity[3]					.134
Hispanic	71/463(15.3%)	38/280(13.6%)	48/257(18.7%)	157/1000(15.7%)	
Black	106/463(22.9%)	82/280(29.3%)	53/257(20.6%)	241/1000(24.1%)	
White	281/463(60.7%)	157/280(56.1%)	150/257(58.4%)	588/1000(58.8%)	
Other	5/463(1.1%)	3/280(1.1%)	6/257(2.3%)	14/1000(1.4%)	
Education					.309
8th grade or less	5/463(1.1%)	3/280(1.1%)	6/257(2.3%)	14/1000(1.4%)	
Some high school, no diploma	38/463(8.2%)	24/280(8.6%)	18/257(7.0%)	80/1000(8.0%)	
High school diploma or GED	110/463(23.8%)	79/280(28.2%)	66/257(25.7%)	255/1000(25.5%)	
Some college or technical training	162/463(35.0%)	95/280(33.9%)	92/257(35.8%)	349/1000(34.9%)	
College degree (such as BA, BS)	123/463(26.6%)	61/280(21.8%)	51/257(19.8%)	235/1000(23.5%)	
Any post college education	25/463(5.4%)	18/280(6.4%)	24/257(9.3%)	67/1000(6.7%)	

[1]Only randomized participants were included.

[2]Fisher exact test was used for Education.

[3]If reported Hispanic=Yes then Ethnicity=Hispanic.

Generated from G:\SAS SUGI\demo_treat.sas on November 13, 2011 at 10:39 PM (Cont'd)

Study XXXXXX

Table 1.0

Demographics by Treatment, overall[1]

	Treatment A (N=463)	Treatment B (N=280)	Treatment AB (N=257)	Overall (N=1000)	p value[2]
Marriage Status					.778
Married	9/463(1.9%)	2/280(0.7%)	4/257(1.6%)	15/1000(1.5%)	
Living with partner/cohabitating	103/463(22.2%)	74/280(26.4%)	61/257(23.7%)	238/1000(23.8%)	
Widowed	231/463(49.9%)	134/280(47.9%)	127/257(49.4%)	492/1000(49.2%)	
Separated/divorced	113/463(24.4%)	68/280(24.3%)	60/257(23.3%)	241/1000(24.1%)	
Single	7/463(1.5%)	2/280(0.7%)	5/257(1.9%)	14/1000(1.4%)	
Annual Income					.627
Missing (no report)	63/463(13.6%)	27/280(9.6%)	44/257(17.1%)	134/1000(13.4%)	
$0 to $5000	63/463(13.6%)	39/280(13.9%)	25/257(9.7%)	127/1000(12.7%)	
$5,001 to $10,000	16/463(3.5%)	8/280(2.9%)	10/257(3.9%)	34/1000(3.4%)	
$10,001 to $20,000	41/463(8.9%)	21/280(7.5%)	29/257(11.3%)	91/1000(9.1%)	
$30,001 to $40,000	65/463(14.0%)	36/280(12.9%)	34/257(13.2%)	135/1000(13.5%)	
$40,001 to $50,000	64/463(13.8%)	39/280(13.9%)	28/257(10.9%)	131/1000(13.1%)	
More than $50,000	151/463(32.6%)	110/280(39.3%)	87/257(33.9%)	348/1000(34.8%)	

[1]Only randomized participants were included.

[2]Fisher exact test was used for Education.

[3]If reported Hispanic=Yes then in ethnicity=Hispanic.

Generated from G:\SAS SUGI\demo_treat.sas on November 13, 2011 at 10:39 PM

Display 2. Summary Table Generated by the ODS Report Writing Interface

The program started with a DATA _null_ statement and had a by processing design variable of “sequence”. The number of table rows in each page was dynamically determined by the difference between designed maximum row lines of 24 and the maximum categories the variable have. A gridded layout was chosen. The code is relatively long, however as we can find, there are several repetitious sections. Each of the repetitions is either between table_start() and table_end() or row_start() and row_end(). These are the objects that define the table title, column header, first line of each variable, the regular table body and the footnote. In practice, we created a set of macros that having one macro for each of the objects. The macros are attached in Appendix 2. After replacing the repetitious sections with the macros, the program has much fewer lines as seen in the following listing:

```
%let dsn= %quote(i:\SAS SUGI\demo_treat.sas);
%let sdate=%sysfunc(putn("&sysdate"d, worddate.));
%let stime=%sysfunc(time(),timeampm9.);
%let pre=trt;
%let width=width=80pct;
%let linemax=24;
%let headname=Treatment A* Treatment B* Treatment AB* Overall;
%let rowhead=ethnicity;
options nodate nonumber orientation=landscape;
ods listing close;
ods rtf file='i:\sas sugi\table 1.0 .rtf' style=sugi_report;
ods escapechar="^";
title j=r'page^{thispage} of ^{lastpage}';
data _null_;
set demo_treat end=eof;
line+1;
by sequence;
if _n_=1 then do;
dcl odsout obj();
obj.layout_gridded(columns:1);
%title;
%colhead(pvalue=yes);
end;
if first.sequence and (&linemax-line) le catmax then do;
obj.table_end();
%foot(lastpage=);
obj.page();
```

```
%title;
%colhead (pvalue=yes);
line=1;
end;
if first.sequence and lowcase(trim(varname))="&rowhead" then do;
%pline1(group=3, overall=yes, pvalue=yes);
end;
else if first.sequence then do;
%pline2(group=3, overall=yes, pvalue=yes);
end;
%data(group=3,overall=yes, pvalue=yes);
if eof then do;
obj.table_end();
%foot(lastpage=yes);
obj.layout_end();
obj.delete();
end;
run;
ods rtf close;
ods listing;
title ;
```

By calling different macros at different palaces, the program works as the master program in this two level structure. This two level structure is adaptive. Any table that has a similar layout , can be created with just some simple modifications of the macros. One very nice feature of the gridded layout is that the column width is adjusted automatically and proportionally, so no more explicit column width estimation is needed if the number of total number of columns is not too large.

GENERATE LISTING

Most of the time, creating listings is easier than creating summary tables, since information is just simply listed out with less manipulation required. However with the power of the ODS Report Writing Interface, we can easily add some dynamic features in the listing. In the next example, the list of participants in one site crosses different pages, so by attaching a "(Con't)" after the site name at the top of the subsequent page, it makes the listing even more friendly to read. The code to create this listing was adopted from the program that creates the summary table, with the two level structure, we can set up the new program with just some minor modifications.



40991	62	Male	Treatment B	86.2	3.6
LA					
30077	61	Male	Treatment B	107.4	3.1
30126	63	Female	Treatment B	90.3	3.4
30196	67	Male	Treatment A	93.8	3.4

Generated from I:\Rui\SAS SUGI\listing.sas on November 14, 2011 at 3:46 PM (Cont'd)



page2 of 5

Study XXXXXX
Listing 1.0
Lab results of participants older than 60 years

Site & Patient ID	Age (Years)	Gender	Treatment	Glucose (mg/dL)	Albumin (g/dL)
LA(Cont'd)					
30380	64	Male	Treatment B	112.5	3.5
30607	63	Male	Treatment A	75.7	2.5
30661	62	Male	Treatment A	109.3	3.0
30683	61	Male	Treatment B	89.2	3.8

Display 3. Sample Listing Generated by the ODS Report Writing Interface

CONCLUSION

ODS Report Writing Interface is a powerful tool to create clinical study reports. The language is object-oriented. It is straightforward and easy to learn. Since every table can be decomposed to different modules, creating a table is a progress of creating the composing objects. After building up the two-level structure, different tables with the similar layout in the same study usually can be very easily created by some minor modification in the macros or macro variables each time.

REFERENCES

1. Daniel O'Conor The power to show: AD Hoc reporting, custom invoices, and form letter. Available at http://support.sas.com/resources/papers/proceedings09/313-2009.pdf

2. Daniel O'Conor.Next Generation data _NULL_ report writing Using ODS OO features. . Available at Http://www2.sas.com/proceedings/sugi28/022-28.pdf

3. David Carr, Lanie Quintana, Akari Yamagami. PROC REPORT or DATA _NULL_: Which to use and how to make it easier. Available at http://www.lexjansen.com/pharmasug/2007/ad/ad16.pdf

4. Richard Koopmann Jr. Experimenting with the ODS DATA step Oject(part I). Available at http://www.lexjansen.com/pnwsug/2007/Richard%20Koopmann%20-%20Experimenting%20with%20the%20ODS%20DATA%20Step%20Object.pdf

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Rui Duan
Department of Epidemiology and Public Health, University of Miami
Clinical Research Building, Room 1071-1
1120 NW 14 Street, Miami, FL 33136
Office: 305-243-4312
Fax: 305-243-5544
Email: rduan@med.miami.edu

Daniel Feaster
Department of Epidemiology and Public Health, University of Miami
Clinical Research Building, Room 1055
1120 NW 14 Street, Miami, FL 33136
Miami, FL 33136
Office: 305-243-7881
Fax: 305-243-5544
Email: dfeaster@biostat.med.miami.edu

Appendix 1.

```
%let dsn= %quote(G:\SAS SUGI\demo_treat.sas);
%let sdate=%sysfunc(putn("&sysdate"d, worddate.));
%let stime=%sysfunc(time(),timeampm9.);
%let pre=trt;
%let width=width=80pct;
%let linemax=24;

options nodate nonumber orientation=landscape;
ods listing close;
ods rtf file='g:\sas sugi\table 1.0 .rtf' style=sugi_report;
ods escapechar="^";
title j=r'page^{thispage} of ^{lastpage}';

data _null_;
set demo_treat end=eof;
line+1;
by sequence;
if _n_=1 then do;
dcl odsout obj();
obj.layout_gridded(columns:1);
obj.table_start(overrides:"&width");
obj.row_start();
obj.format_cell(data: "Study XXXXXX",Overrides:'font_size=12pt
font_weight=bold',inhibit:"tbrl");
obj.row_end();
obj.row_start();
obj.format_cell(data: "Table 1.0",Overrides:'font_size=12pt
font_weight=bold');
obj.row_end();
obj.row_start();
obj.format_cell(data:"Demographics by Treatment, overall^{super
1}",Overrides:'font_size=12pt font_weight=bold');
obj.row_end();
obj.table_end();
obj.table_start(overrides:"&width");
obj.head_start();
obj.row_start();
obj.format_cell(data:"",overrides:'width=0.1in');
obj.format_cell(data:"");
obj.format_cell(data:"Treatment A*(N=&trt_1)",split:'*');
obj.format_cell(data:"Treatment B*(N=&trt_2)",split:'*');
obj.format_cell(data:"Treatment AB*(N=&trt_3)",split:'*');
obj.format_cell(data:"Overall*(N=&trt_4 )",split:'*');
obj.format_cell(data:"p value^{super 2}");
obj.row_end();
obj.head_end();
end;
if first.sequence and (&linemax-line) le catmax then do;
obj.table_end();
obj.table_start(overrides:"&width");
obj.row_start();
obj.format_cell(data: '^{super 1}Only randomized participants were
included.', overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
```

```
obj.format_cell(data: '^{super 2}Fisher exact test was used for
Education.',overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data: '^{super 3}If reported Hispanic=Yes then
Ethnicity=Hispanic.',overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data:"Generated from &dsn on &sdate at
&stime^R/rtf'\tab\tab'(Cont'd)", overrides: 'just=L
font_size=9pt',inhibit:"brl");
obj.row_end();
obj.table_end();

obj.page();

obj.table_start(overrides:"&width");
obj.row_start();
obj.format_cell(data: "Study XXXXXX",Overrides:'font_size=12pt
font_weight=bold',inhibit:"tbrl");
obj.row_end();
obj.row_start();
obj.format_cell(data: "Table 1.0",Overrides:'font_size=12pt
font_weight=bold');
obj.row_end();
obj.row_start();
obj.format_cell(data:"Demographics by Treatment, overall^{super
1}",Overrides:'font_size=12pt font_weight=bold');
obj.row_end();
obj.table_end();
obj.table_start(overrides:"&width");
obj.head_start();
obj.row_start();
obj.format_cell(data:"",overrides:'width=0.1in');
obj.format_cell(data:"");
obj.format_cell(data:"Treatment A*(N=&trt_1)",split:'*');
obj.format_cell(data:"Treatment B*(N=&trt_2)",split:'*');
obj.format_cell(data:"Treatment AB*(N=&trt_3)",split:'*');
obj.format_cell(data:"Overall*(N=&trt_4 )",split:'*');
obj.format_cell(data:"p value^{super 2}");
obj.row_end();
obj.head_end();
line=1;
end;
if first.sequence and lowcase(trim(varname))='ethnicity' then do;
obj.row_start();
obj.cell_start(column_span: 6);
obj.format_text(text: 'Ethnicity^{super 3}',overrides: 'just=L');
obj.cell_end();
obj.format_cell(data: pvalue);
obj.row_end();
end;
else if first.sequence then do;
obj.row_start();
obj.cell_start(column_span: 6);
obj.format_text(text:varname,overrides: 'just=left');
obj.cell_end();
```

```
obj.format_cell(data: pvalue);
obj.row_end();
end;
obj.row_start();
obj.format_cell(data:'',overrides: 'width=0.1in');
obj.format_cell(data:statlist,overrides: 'just=L');
obj.format_cell(data: trt_1);
obj.format_cell(data: trt_2);
obj.format_cell(data: trt_3);
obj.format_cell(data: trt_4);
obj.format_cell(data: '');
obj.row_end();
if eof then do;
obj.table_end();
obj.table_start(overrides:"&width");
obj.row_start();
obj.format_cell(data: '^{super 1}Only randomized participants were
included.', overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data: '^{super 2}Fisher exact test was used for
Education.',overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data: '^{super 3}If reported Hispanic=Yes then in
ethnicity=Hispanic.',overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data:"Generated from &dsn on &sdate at &stime", overrides:
'just=L font_size=9pt',inhibit:"brl");
obj.row_end();
obj.table_end();
obj.layout_end();
obj.delete();
end;
run;
ods rtf close;
ods listing;
title ;
```

Appendix 2

```
%Macro title;
obj.table_start(overrides:"&width");
obj.row_start();
obj.format_cell(data: "Study XXXXXX",Overrides:'font_size=12pt
font_weight=bold',inhibit:"tbrl");
obj.row_end();
obj.row_start();
obj.format_cell(data: "Table 1.0",Overrides:'font_size=12pt
font_weight=bold');
obj.row_end();
obj.row_start();
obj.format_cell(data:"Demographics by Treatment, overall^{super
1}",Overrides:'font_size=12pt font_weight=bold');
obj.row_end();
obj.table_end();
obj.table_start(overrides:"&width");
%mend;

%macro colhead (pvalue=);
obj.table_start(overrides:"&width");
obj.head_start();
obj.row_start();
obj.format_cell(data:"",overrides:'width=0.1in');
obj.format_cell(data:"");
%let i=1;
%let charlist=*;
%do %until(%scan(&headname, %eval(&i),*) eq );
%let name=%scan(&headname, %eval(&i),*);
obj.format_cell(data:"&name*(N=&&trt_&i)",split:'*');
%let i=%eval(&i+1);
%end;

%if &Pvalue ne  %then %do;
obj.format_cell(data:"p value^{super 2}");
%end;
obj.row_end();
obj.head_end();
%mend;

%macro pline2(group=, overall=, pvalue=);
obj.row_start();
%if  &overall eq  and &pvalue eq  %then %do;
obj.cell_start(column_span: %eval(2+&group));
obj.format_text(text: varname,overrides: 'just=left');
obj.cell_end();
obj.row_end();
%end;
%else %if &overall eq  %then %do;
obj.cell_start(column_span: %eval(2+&group));
obj.format_text(text: varname,overrides: 'just=left');
obj.cell_end();
obj.format_cell(data: pvalue);
obj.row_end();
%end;
%else %if &pvalue eq  %then %do;
```

```
obj.cell_start(column_span: %eval(2+&group));
obj.format_text(text: varname,overrides: 'just=left');
obj.cell_end();
obj.row_end();
%end;
%else %do;
obj.row_start();
obj.cell_start(column_span: %eval(3+&group));
obj.format_text(text: varname,overrides: 'just=left');
obj.cell_end();
obj.format_cell(data: pvalue);
obj.row_end();
%end;
%mend;

%macro pline1(group=, overall=, pvalue=);
obj.row_start();
%if  &overall eq  and &pvalue eq  %then %do;
obj.cell_start(column_span: %eval(2+&group));
obj.format_text(text:trim(varname)||"^{super3}",overrides: 'just=left');
obj.cell_end();
obj.row_end();
%end;
%else %if &overall eq  %then %do;
obj.cell_start(column_span: %eval(2+&group));
obj.format_text(text:trim(varname)||"^{super 3}",overrides: 'just=left');
obj.cell_end();
obj.format_cell(data: pvalue, format:'$pvalue');
obj.row_end();
%end;
%else %if &pvalue eq  %then %do;
obj.cell_start(column_span: %eval(2+&group));
obj.format_text(text:trim(varname)||"^{super 3}",overrides: 'just=left');
obj.cell_end();
obj.row_end();
%end;
%else %do;
obj.row_start();
obj.cell_start(column_span: %eval(3+&group));
obj.format_text(text:trim(varname)||"^{super 3}",overrides: 'just=left');
obj.cell_end();
obj.format_cell(data: pvalue);
obj.row_end();
%end;
%mend;

%macro data(group=,overall=, pvalue=);
obj.row_start();
obj.format_cell(data:'',overrides: 'width=0.1in');
obj.format_cell(data: statlist,overrides: 'just=L');
%do i=1 %to %eval(&group);
obj.format_cell(data:&pre._&i);
%end;
%if &overall ne  %then %do;
obj.format_cell(data:&pre._%eval(&group+1));
%end;
%if &pvalue ne  %then %do;
```

```
obj.format_cell(data: '');
%end;
obj.row_end();
%mend;

%Macro foot(lastpage=);
obj.table_start(overrides:"&width");
obj.row_start();
obj.format_cell(data: '^{super 1}Only randomized participants were
included.', overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data: '^{super 2}Fisher exact test was used for
Education.',overrides: 'just=L font_size=9pt');
obj.row_end();
obj.row_start();
obj.format_cell(data: '^{super 3}If reported Hispanic=Yes then
Ethnicity=Hispanic.',overrides: 'just=L font_size=9pt');
obj.row_end();
%if &lastpage ne %then %do;
obj.row_start();
obj.format_cell(data:"Generated from &dsn on &sdate at &stime", overrides:
'just=L font_size=9pt',inhibit:"brl");
obj.row_end();
obj.table_end();
%end;
%else %do;
obj.row_start();
obj.format_cell(data:"Generated from &dsn on &sdate at
&stime^R/rtf'\tab\tab'(Cont'd)", overrides: 'just=L
font_size=9pt',inhibit:"brl");
obj.row_end();
obj.table_end();
%end;
%mend;
```